

UM MODELO DE KALIMBA EM CSOUND USADO EM O LIVRO DOS SONS

A TYPE OF KALIMBA IN CSOUND USED IN THE BOOK OF SOUNDS

Rodolfo Coelho de Souza - FFCL de Ribeirão Preto/USP
rcoelho@usp.br

Resumo: O objetivo desta pesquisa foi projetar em Csound uma emulação simples de kalimba que pudesse ser usada na composição de *O Livro dos Sons* para orquestra e sons eletrônicos, servindo também como primeiro estágio de uma modelagem física completa do instrumento. O modelo proposto, inspirado em Välimäki (1996) e Serra (1997), combina princípios básicos de modelagem de instrumento e de seu espectro, através de síntese aditiva. O projeto partiu da análise espectral de uma amostra que forneceu funções de mapeamento entre a fonte de excitação e o ressonador. Atentou-se, no modelo, para a prevalência de parciais inarmônicos, variações no tempo da amplitude dos parciais e de componentes de ruído, considerados teórica e experimentalmente. Propôs-se também uma afinação especial das teclas da kalimba, concebida a partir dos resultados da análise espectral, para se conferir um caráter idiomático ao instrumento virtual.

Palavras-chave: Modelagem de instrumento; Síntese sonora; Composição eletroacústica.

Abstract: The goal of this research was to design in Csound a simple kalimba emulator that might be used in a composition named *The Book of Sounds* for orchestra and electronic sounds and also further expanded pursuing a complete physical model of the instrument. The proposed model combined basic principles of instrument and spectral modeling, using additive synthesis. The design process started with the spectral analysis of a sample that yielded mapping functions between the exciter and the resonator. Aspects of prevailing inharmonic partials, amplitude variation on time of partials and noise components were theoretically and experimentally approached in the model. A particular tuning of the kalimba keys, conceived after the results of the spectral analysis, was proposed in order to accomplish an idiomatic character for the virtual instrument.

Keywords: Instrument modeling; Sound synthesis; Electroacoustic composition.

1. Introdução ao problema da modelagem física de instrumentos

A imitação de sons de instrumentos acústicos naturais por meio de algoritmos computacionais constitui uma linha de pesquisa chamada de ‘modelagem física’ na medida em que ela se opõe à categoria de ‘modelagem por processamento de sinal’ na qual os sons são gerados por modelos matemáticos abstratos (De Poli 1998, p. 62). Certamente, no final, ambas as abordagens resultam num algoritmo computacional, contudo, no caso do modelo físico, podemos relacionar em nossas mentes as variações do resultado sonoro a gestos imaginários ou a escolhas de parâmetros moduláveis que faria um músico virtual ao executar o instrumento.

Uma vez que há uma sobreposição entre os modelos físicos e de processamento de sinal, diferentes autores podem incluir ou não certas técnicas como parte da categoria ‘modelagem física’. De Poli (1998, p. 65-67) descreve três abordagens gerais: modelos celulares, modelos baseados em diferenças finitas e modelos de onda. Modelos de diferenças finitas e de onda diferem em que o de diferenças finitas encontra valores discretos para a equação de onda, enquanto modelos de onda (*wave-guide*) trabalham com discretizações da solução da equação de onda. Ao lado destes, Serra (1997, p. 90) descreve outra abordagem geral que ele chama de ‘modelo espectral’. O modelo espectral de Serra é uma combinação de síntese aditiva, replicando os parciais senoidais do som natural, somados a certos componentes de ruído considerados como resíduo. Este modelo é particularmente eficiente para sons que contém parciais inarmônicos ou parcelas significativas de ruído, o que é precisamente o que ocorre no som da kalimba que tem espectro inarmônico e componentes ruidosos. Este método de busca de frequências, estendido por Smith e Serra para aplicações musicais (Serra 1997), “adota a posição de que um sinal sonoro é composto por coleções de senóides com frequências arbitrárias, isto é, não existem proporções particulares entre as frequências, e que cada componente de frequência não necessariamente existe durante toda a duração do sinal, podendo até mesmo conter componentes de frequência que tem existências extremamente curtas, tipicamente aquelas que formam *clusters* que imitam impulsos de ruídos” (Beauchamp 2007, p. 26).

Após alguns experimentos com as diferentes abordagens, percebemos que os métodos clássicos de modelagem física ofereceriam no nosso caso muitas dificuldades para se implementar em Csound, pois exigiriam a codificação de novos operadores. Por outro lado concluímos que era possível adotar uma aproximação simples e rápida do método de modelagem espectral de Serra que nos fornecesse resultados razoavelmente parecidos com o som que estávamos tentando simular. Contudo é importante enfatizar que nossa abordagem não foi plenamente consistente com o método de Serra, uma vez que não seguimos até as últimas consequências os complexos aspectos do método de análise-síntese que ele propõe. Para nossos propósitos a simplificação adotada foi satisfatória, uma vez que nosso objetivo não era sintetizar sons de kalimba com perfeição absoluta, mas sim projetar um instrumento em Csound que produzisse resultados com similarida-

de auditiva aceitável à do instrumento acústico, e ao mesmo tempo, definiu uma quantidade restrita de parâmetros de controle que pudessem ser facilmente operados virtualmente para gerar novos e interessantes sons, os quais poderiam ser usados na composição da obra a que demos o nome de *O Livro dos Sons*.

Tomando o método de Serra como um conceito genérico de “combinação de síntese aditiva para os componentes senoidais e de síntese subtrativa para a parcela de ruído” (Serra 1997, p. 92), começamos analisando amostras de sons de kalimba para detectar quais os parciais presentes e estudar a variação de suas amplitudes no tempo. O passo seguinte foi subtrair, com filtros, estes parciais da amostra original. O resíduo remanescente foi então simulado através da filtragem de um sinal de ruído branco.

2. Características da kalimba

A kalimba é um instrumento de origem africana que se difundiu nas Américas devido à imigração de escravos africanos, principalmente no Caribe e na costa do Brasil. Não existe um padrão fixo para a construção do instrumento, entretanto todas as kalimbas terão como fonte sonora um número variável de teclas feitas de lâminas de metal ou madeira e que são dispostas em padrões alternados e não-lineares. Essas teclas são fixadas num único lado, ao contrário do que ocorre com os instrumentos de corda, e transmitem a vibração para um cavalete que é fixado na tampa de um ressonador. O número de teclas ou lâminas pode variar consideravelmente. Elas são tocadas pelos polegares das duas mãos que ao mesmo tempo seguram o instrumento usando os outros dedos. Devido a esse mecanismo de acionamento, as kalimbas são também conhecidas como pianos de polegar. As vibrações das teclas são amplificadas pelo corpo do instrumento, que funciona como um ressonador de Helmholtz e pode ter muitas formas. As formas mais usuais são as de caixa prismática vazia ou de uma cabaça de madeira colada a uma tampa de pouca espessura, também de madeira, que entra em ressonância com as vibrações transmitidas pelas teclas.

A performance da kalimba permite algumas variações nas características do som produzido. Quando o polegar toca a lâmina metálica, por um breve instante há uma fricção entre as superfícies do dedo e do metal,

causando um impulso vibratório inicial, que é composto principalmente de ruído, mas que já contém um sinal perceptível da frequência fundamental da lâmina. Assim que o polegar perde contato com o metal, a lâmina começa a vibrar livremente, seguindo o modelo de barras pinçadas fixadas de um só lado, como detalharemos adiante.

A energia transferida pelo polegar pode variar de amplitude, mas entre limites não muito extensos, todavia com diferenças suficientes para conferir características de variação dinâmica e mudanças significativas no impulso transiente, permitindo ao instrumentista conferir alguma variedade de coloração de timbre à execução da kalimba. A vibração total é amortecida porque a lâmina, engastada no cavalete, transmite a energia para o tampo e dele para o ressonador, fazendo a energia vibratória se dissipar rapidamente pela emissão da energia sonora. O ressonador captura e amplifica a amplitude da vibração da lâmina pinçada pelo polegar, portanto a envoltória de amplitudes não deve ser um mero decaimento exponencial, como seria característico de uma lâmina engastada num corpo de massa infinita. As amostras sonoras estudadas confirmam essa hipótese, demonstrando que o envelope de amplitude é uma curva composta e complexa que se pode simular com um mapeamento do funcionamento exibido pelas amostras sonoras.

Note-se também que, assim que a energia de uma tecla é transmitida para o cavalete, ela poderia provocar não só a vibração do ressonador, mas também das teclas adjacentes. Ou seja, a vibração do ressonador causaria uma retro-alimentação de energia nas barras. O efeito geral seria similar ao da ressonância de uma harpa, ou de um piano com o pedal abafador abaixado. A quantidade de energia transferida pelo cavalete às outras teclas depende das características de construção de cada instrumento, mas podemos assumir que o coeficiente de transferência é pequeno porque as respectivas frequências naturais de ressonância não são relacionadas harmonicamente. Retiramos do algoritmo em Csound apresentado neste artigo, o trecho que simulava o efeito de *feed-back*, porque ele resultou num código extenso que, entretanto, produzia variações pequenas no resultado auditivo. Apesar de ser um efeito sutil de segunda ordem, a ressonância composta entre teclas e ressonador é um componente conceitual interessante da kalimba que foi de fato incorporada à versão mais extensa do algoritmo usado em *O Livro dos Sons*, uma vez que sua manipulação, sem

a preocupação de produzir resultados realistas de simulação da kalimba, produziu resultantes sonoras originais do ponto de vista composicional.

Portanto nosso modelo físico, também inspirado em Välimäki (1996, p. 82, Figura 4), foi construído com três partes: a fonte de excitação, o ressonador e um bloco intermediário que mapeia a troca de informações entre as outras partes (ver Figura 1). Observe-se que o bloco intermediário incorpora o efeito físico de retro-alimentação através de um mapeamento por dados virtuais: o que no mundo real acontece como uma complexa interação entre massas, das teclas e do ressonador, é simulado por algumas funções paramétricas.



Figura 1: Componentes básicos do modelo instrumental da kalimba.

3. Sobre a física da tecla de kalimba

As laminas metálicas da kalimba são percutidas com os polegares na sua extremidade livre, enquanto o lado oposto está engastado no cavalete de um tampo de uma caixa de ressonância. A vibração da tecla pode ser descrita com boa aproximação pela equação de vibração de uma barra longa e fina, perfeitamente elástica.

A equação geral para as ondas de flexão em barras elásticas é (Fletcher 1998, p. 60-63):

$$y = \cos(\omega t + \phi) [A \cosh kx + B \sinh kx + C \cos kx + D \sin kx]. \text{ (eq. 1.1)}$$

em que A , B , C e D são quatro constantes arbitrárias que podem ser determinadas após a escolha de quatro condições de contorno, duas para cada extremidade da barra.

Para o caso de uma barra de comprimento L , engastada em $x = 0$ e livre em $x = L$, as condições de contorno implicam que:

$$A + C = B + D = 0$$

e a solução para a equação transcendental derivada de quarta ordem 1.1 produz as seguintes frequências resultantes

$$f_n = \frac{\pi K}{8L^2} \sqrt{\frac{E}{\rho} [1.194^2, 2.988^2, 5^2, \dots, (2n - 1)^2]}. \text{ (eq. 1.2)}$$

Para uma dada barra de comprimento L , construída com uma determinada geometria, com um material que tenha as constantes K (constante de mola), E (módulo de Young) e ρ (densidade), as variáveis da equação 1.2 se tornam fixas e podemos calcular as relações entre as frequências naturais da barra:

$$f_2 = 6.27 f_1, f_3 = 17.55 f_1, f_4 = 35.39 f_1, \text{ e assim por diante.}$$

Este resultado traz notáveis conseqüências para qualquer instrumento construído com este tipo de barra percutida. De modo muito diferente do comportamento das cordas vibratórias ou do ar que vibra em tubos longos, as frequências geradas por barras engastadas em um só lado não apresentam nenhuma relação direta com os harmônicos naturais. Mais ainda, não se trata de um caso de espectro harmônico que inclua frequências inarmônicas, tal como acontece com o comportamento de sinos. O espectro gerado por esse tipo de tecla é totalmente inarmônico. A frequência fundamental é bem mais grave e distante das frequências superiores que rapidamente crescem em direção aos registros mais agudos. A frequência fundamental é consideravelmente mais intensa do que os outros parciais, de modo que o som é percebido com uma altura definida, apesar de que, devido a sua inarmonicidade intrínseca, a qualidade do som aproxima-se muito da família da percussão, particularmente de xilofones e tambores de madeira. Portanto pode-se concluir que não é uma mera coincidência que as culturas musicais africanas e afro-brasileiras, que tem uma forte base na linguagem percussiva, tivessem preferência pelo uso deste dispositivo como gerador sonoro de seus instrumentos musicais, enquanto eles são raramente (ou jamais) empregados nos instrumentos ocidentais tradicionais. Esse fato já havia chamado a atenção de Stedman em 1777 (apud Frungillo, 2002) que noticia a frequência com que o instrumento era encontrado no Rio de Janeiro naquela época.

4. Análise de uma amostra sonora de kalimba

A amostra de kalimba que usamos foi gravada com um instrumento comprado num mercado popular de Salvador, Bahia. Certamente poderíamos ter usado um instrumento de melhor luteria, mas estávamos interessados na amostra de um instrumento rústico que trouxesse características da tradição folclórica. Para fazer a análise e rastreamento das frequências utilizamos a função **lpanal** do programa Csound, com uma extensão de 500 janelas (Klapper 2000, p. 528). Numa etapa posterior desta pesquisa deveremos utilizar um processo mais sofisticado para a análise e rastreamento de frequências, pois a função de Csound que utilizamos não é das mais sofisticadas para estudar esse tipo de amostra. Ainda assim a análise espectral apresentou resultados que consideramos satisfatórios porque eles concordaram com as expectativas do modelo teórico.

A análise espectral detectou duas parciais principais com energia significativa consistente com o do modelo teórico de barra engastada em um lado: a fundamental, centrada em 259 Hz, e uma segunda parcial a 1648 Hz, com uma discrepância de 1.5% da frequência prevista como 1624 Hz pelo modelo teórico, mas ainda assim numa relação totalmente inarmônica. Este desvio pode ter sido causado por irregularidades geométricas e elásticas da barra ou talvez um deslocamento de frequências no programa de rastreamento de alturas (como veremos, o resultado poderia implicar em relações harmônicas com uma hipotética “fundamental ausente”). Parciais superiores previstos nas frequências de 4545 Hz e 9166 Hz podem ter estado presentes, mas com energias muito baixas, que o espectrograma não foi capaz de confirmar. Devemos mencionar ainda que nesta fase da análise excluiu-se a fase transiente inicial da amostra porque seu caráter ruidoso provocava grande dispersão de resultados. Por outro lado a análise revelou diversos picos de menor intensidade ao redor da frequência fundamental, acima e abaixo dela. Foi simples deduzir-se que o ressonador era o responsável por esses parciais, porque bastou bloquear a abertura do ressonador ao se gravar a amostra, para que a intensidade dessas frequências fosse substancialmente reduzida. A Figura 2 apresenta os picos das parciais presentes na análise espectral.




Figura 2: Análise espectral de uma amostra de kalimba: picos dos parciais presentes.

Observe-se que a Figura 2 poderia conduzir a uma interpretação enganosa se entendêssemos a representação dos parciais como sendo componentes fixos estáveis, o que definitivamente não é correto. De fato **lpa-nal** fornece as amplitudes variáveis dos parciais, mas uma demonstração gráfica de seus resultados seria bastante complexa de produzir e sua apresentação em forma de tabela seria tediosa e extensa. Após uma análise detalhada dos resultados chegamos à conclusão de que podemos distinguir três comportamentos básicos distintos para os parciais: a fundamental tem uma presença que se sustenta ao longo de toda a amostra e sua envoltória dinâmica pode ser simulada por uma combinação de controles ADSR e exponenciais; o segundo parcial inarmônico, predito pelo modelo de barra engastada, tem uma duração mais curta, algo em torno de metade do tempo total da amostra, com uma envoltória de amplitude próxima de um decaimento exponencial; finalmente, todos os outros parciais, basicamente gerados pelo ressonador, permaneciam ao longo de toda a duração e exibiam um decaimento aproximadamente exponencial, porém sem o incremento de sustentação não exponencial exibido pela fundamental.

Ao tentar simular os componentes de frequência que apareceram ao redor da fundamental, percebemos que eles poderiam ser considerados como formantes do ressonador de Helmholtz respondendo às frequências de 259 Hz e 1648 Hz da fonte de excitação. Há, de fato, coincidências notáveis entre as frequências medidas e o que poderia ser interpretado como sendo harmônicos de uma fundamental ausente que resultariam das frequências de excitação divididas por números inteiros. A Tabela 1 demonstra esse modelo. Como vemos, há quatro frequências de ressonância que se relacionam com o segundo parcial de excitação de 1648 Hz, como se este

fosse o 70º parcial de uma fundamental ausente com 23,5 Hz. Coincidentemente, um deles, o 11º harmônico, seria também a outra frequência de excitação com 259 Hz. De modo similar, as outras frequências de ressonância podem ser interpretadas como harmônicos de fundamentais ausentes a 28,8 Hz, 37 Hz e 43,2 Hz, que teriam a frequência de excitação de 259 Hz respectivamente como 9º, 7º, e 6º harmônicos.

Fundamental	Parciais												
N. Parcial → 23.5 = 1648/11		5 118	6 141		8 188			11 259				70 1648	
N. Parcial → 28.8 = 259/9						7 202		8 230	9 259		11 317		
N. Parcial → 37 = 259/7	3 111							7 259	8 296			9 333	
N. Parcial → 43.2 = 259/6				4 173			5 216		6 259				
Medição, em Hz	112	120	141	174	190	202	215	231	259	297	318	332	1648

Tabela 1: Modelo de mapeamento da retro-alimentação do ressonador.

Pode ser que seja mais do que uma mera coincidência que os inteiros 6, 7, 9 e 11, usados para se obter as fundamentais ausentes da Figura 2, sejam muito similares aos inteiros 5, 7, 9 e 11 que aparecem na Equação 1.2 como coeficientes dos expoentes quadráticos. Contudo não fomos capazes de imaginar uma explicação convincente para essa semelhança. Seja como for, as diferenças constatadas entre as frequências medidas e aquelas propostas pelo modelo de mapeamento são negligenciáveis. Portanto, se este modelo não fornece uma explicação teórica para a causalidade física, ele fornece um conjunto de funções de mapeamento que, com razoável precisão, simulam a retro-alimentação do ressonador conforme o modelo proposto na Figura 1.

O resíduo de ruído foi certamente o aspecto mais difícil de tratar. Fazendo experiências com as amostras gravadas, percebemos que, após filtrar os parciais, uma parte substancial do resíduo parecia não ter nenhuma relação com os demais componentes do som. De algum modo esse resíduo tingia a amostra com uma pátina de ruído amortecido que reduzia o brilho da amostra e principalmente prejudicava suas qualidades musicais, sem agregar nenhum caráter musical agradável. Decidimos por isso simplesmente desconsiderar essa parte do resíduo, conservando apenas o transiente inicial, modelado com um impulso padrão de ruído branco com decai-

mento exponencial. Para a próxima etapa da pesquisa, em que se pretende melhorar a semelhança entre o som sintetizado e a amostra gravada, esta será certamente uma parte do modelo que pode ser melhorada.

5. Uma abordagem espectral para a execução da kalimba

Uma segunda parte do nosso problema de modelagem é concernente ao excitador. Na Figura 1 mostramos que ele depende de três componentes: o impulso do polegar, a resposta da tecla e os dados de entrada do executante. Todavia não apresentamos ali detalhes específicos da interação desses componentes. Na Figura 3 elaboramos a contribuição desses componentes interpretando sua interação através de parâmetros de retardo. O impulso do polegar pode atrasar o início da vibração da tecla por um tempo variável, dentro de certa faixa de variação. Este efeito foi representado pelo parâmetro de retardo 1. O já mencionado efeito de vibrações por simpatia, causado pela vibração da tecla, foi simulado por um filtro de pente aplicado à vibração de excitação. Embora esta não seja uma representação exata daquele efeito complexo, tal expressão foi capaz de fornecer uma simulação simples e razoavelmente precisa daquele efeito. Este efeito secundário também é afetado por outro tempo de retardo, modelado como sendo duas vezes o retardo 1.




Figura 3: Modelo para a interação interna do excitador.

Como mostramos na modelagem do instrumento, o próprio som da kalimba é influenciado pela afinação de suas teclas devido às múltiplas ressonâncias de retro-alimentação. A kalimba de que tiramos a amostra deste estudo tinha 7 teclas, mas Nketia (1974) e Agordoh (2005) esclarecem que não existe um padrão universal para sua afinação, uma vez que as kalimbas africanas tradicionais não tem sequer um número padrão de teclas.

Encontramos um site de internet dedicado ao instrumento (ver ‘Kalimba site’ nas referências) que apresenta algumas das diversas afinações possíveis, variando entre modelos pentatônicos, diatônicos e de tons-inteiros. As kalimbas africanas das tribos nativas são descritas como tendo microafinações e relações inarmônicas aleatórias entre as teclas. Mas de modo geral, as notas ascendem alternadamente entre a mão esquerda e mão direita porque o executante também alterna os acionamentos dos polegares.




Figura 4: Possível disposição de notas em uma kalimba de sete teclas.

Nosso modelo propõe uma abordagem espectral para a afinação da kalimba virtual, assinalando para a fundamental de cada tecla uma das frequências encontrada na análise espectral. A fundamental de 259 Hz foi assinalada à tecla 3 para que se obtivesse uma disposição similar à das frequências medidas, tal como exibido na Tabela 2. Observe-se que não existe aqui o princípio de equivalência de oitavas, nem os intervalos seguem um padrão regular, embora eles gravitem em torno de uma proporção de 1,1225 que é o intervalo temperado de tom inteiro, o que pode ser uma indicação de uma sensibilidade espectral na afinação das kalimbas das populações nativas, tal como reportado em resultados medidos por Agordoh (2005).

tecla	1	2	3	4	5	6	7
Freq. [Hz]	173	188	216	230	259	296	333
f_n/f_{n-1}	-	1.0867	1.1489	1.0648	1.1261	1.1429	1.125

Tabela 2: Afinação de uma kalimba virtual com abordagem espectral.

Para demonstrar o potencial musical do modelo, apresentamos a seguir uma implementação simplificada em Csound, que executa uma frase típica de kalimba (no que se refere à alternância de polegares) conforme mostrado na Figura 5. Essa frase faz parte dos sons eletrônicos usados em *O Livro dos Sons*, obra encomendada pela OSESP - Orquestra Sinfônica do Estado de São Paulo para a temporada regular de concertos de 2010 e para a abertura do Festival de Inverno de Campos do Jordão daquele ano.




Figura 5: Exemplo de uma frase idiomática de kalimba.

6. Implementação em Csound

```

; KALIMBA.orc
; kalimba sound generated by spectrum and instrument model
sr      =      44100
kr      =      4410
ksmps   =      10
nchnls  =      1
        instr  1
idur    =      p3
iamp    =      10000*p4                ; impulse factor: range 1-2
ipch    table  p5, 7
idel1   =      p6                      ; delay1: range 1-5
; thumb-noise impulse
kenv8   expseg.00001,.01, iamp/4,.06,.00001
asig8   rand    kenv8, 1
; fundamental frequency with resonance reinforcement
kenv1   linseg    iamp, idur*.1, iamp/5, idur*.3, iamp/20, idur*.6, 0
asig1   oscili    kenv1, ipch, 1
asig1   delay    asig1, 0.0025*idel1                ; delay after thumb impulse
; harmonic and inharmonic components
kenv2   expon    iamp/6, idur, 0.0001
kenv5   expon    iamp/2, idur*.5, 0.0001
asig2   oscili    kenv2, ipch/6, 2                  ; 6th vibration mode
asig3   oscili    kenv2, ipch/7, 3                  ; 7th vibration mode
asig4   oscili    kenv2, ipch/9, 4                  ; 9th vibration mode
asig5   oscili    kenv5, ipch/11, 5                 ; main inharmonic
asig6   oscili    kenv5, ipch/11, 6
asig7   =      asig1+asig2+asig3+asig4+asig5+asig6 ; adding components
; sympathetic vibrations
kbs     linen    iamp/100, 0.002, idur, 0.998
asigbs  oscili    kbs, ipch, 1
arvb    comb     asigbs, idur/5, 0.004
arvb    delay    arvb, 0.005*idel1                ; delay after thumb impulse
asig    =      (asig7+arvb+asig8)/1.15            ; adding and normalizing
        out     asig
        endin

```

```

; KALIMBA.sco
f1 0 1024 10 1 ; main partial
f2 0 1024 10 0 0 0 .08 .18
f3 0 1024 10 0 0 0 .08 0 0 0 0 .18 .08
f4 0 1024 10 0 0 0 0 0 0 0 .18 .3 0 0 .18
f5 0 1024 9 70 .32 0 ; 2nd partial
f6 0 1024 10 0 0 0 0 .08 .08 0 .08
f7 0 8 -2 0 173 188 216 230 259 296 333 ;tunning
; nn start dur amp N-key delayfac pch
; range: 1-2 1-7 1-5 Hz
; isolated note, similar to the sampled sound
i01 0.15 2 2 5 3 ; 259
s
; example of kalimba phrase
i01 0.15 2 2.0 7 1 ; 333
i01 0.3 . 1.7 1 5 ; 173
i01 0.45 . 1.4 3 4 ; 216
i01 0.6 . 1.0 4 3 ; 230
i01 0.75 . 1.3 5 2 ; 259
i01 0.9 . 1.6 2 1 ; 188
i01 1.05 . 2.0 3 5 ; 216
i01 1.2 . 1.7 6 4 ; 296
i01 1.35 . 1.4 1 3 ; 173
i01 1.5 . 1.0 4 2 ; 230
i01 1.65 . 1.3 5 1 ; 259
i01 1.8 . 1.6 2 5 ; 188
i01 1.95 . 2.0 7 4 ; 333
i01 2.1 . 1.7 1 3 ; 173
i01 2.25 . 1.4 3 2 ; 216
i01 2.4 . 1.0 4 1 ; 230
i01 2.55 . 1.3 5 5 ; 259
i01 2.7 . 1.6 2 4 ; 188
i01 2.85 . 2 3 3 ; 216
i01 3.00 . 1.7 6 2 ; 296
i01 3.15 . 1.4 1 1 ; 173
i01 3.3 . 1.0 4 5 ; 235
i01 3.45 . 1.3 5 4 ; 259
i01 3.6 . 1.6 2 3 ; 188
e

```

The image displays a page from a musical score for the piece "O Livro dos Sons". The score is arranged in a standard orchestral format with multiple staves. At the top, the woodwind section includes Oboe (Ob.), English Horn (E. Hn.), Bassoon (Bsn.), and Contrabass (C. Bn.). The brass section includes Horns I-IV (Hn. I-IV), Trumpets (B. Tpt.), Trombones I-II (Tbn. I-II), and Tuba (B. Tbn. Tuba). The percussion section (Perc.) includes Wind chimes, shell, glass, Guitar, and Xylophone. The Harp (Hrp.) part is also present. A specific section is dedicated to "FILE 02 KALIMBA sounds en dehors", which includes a diagram of a kalimba instrument and a description: "P with randomly floating harmonics and modulated amplitude (only approximately synchronized)". The score includes various musical notations such as dynamics (p, f, mf, mp, pp), articulation (sord., senza sord.), and performance instructions. The key signature is one sharp (F#) and the time signature is common time (C).

Figura 6: Trecho da partitura de *O Livro dos Sons* que emprega a frase de kalimba.

Comentários finais

Como nota pessoal, relembro um encontro memorável que tive num congresso internacional com um músico de Moçambique, executante de kalimba, cuja performance do instrumento encantou a platéia do evento. Apesar de que isto aconteceu há muitos anos atrás, por certo foi a memória daquele encontro que me forneceu uma parte importante da inspiração para esta pesquisa. Naquele mesmo evento o músico africano ouviu uma de minhas composições para percussão. Após o concerto ele me procurou para dizer que, dentre todas as peças que ele ouvira naquele congresso, a minha era a única que tinha *m'bira*. Obviamente ele tinha encontrado na minha peça algum elo entre a música africana e a brasileira, o que não devia me surpreender. Contudo ele descreveu essa conexão de um modo peculiar. O instrumento que chamamos kalimba, ele o chama de *m'bira*. Mas quando ele se referiu à minha peça usando o termo *m'bira*, ele não se referia ao instrumento em si, porque minha peça não empregava uma kalimba. Ele estava reconhecendo que a linguagem empregada naquela composição, minimalista em certa medida, tinha criado uma aura de ressonância que lhe era familiar. Esta era a *m'bira* a que ele se referia, talvez certa “alma” de kalimba que ele ouvira em minha peça.

O projeto de síntese da kalimba que estamos desenvolvendo tinha entre seus objetivos a reprodução daquela aura. Os aspectos óbvios que podiam contribuir para alcançar esse objetivo eram o efeito do ressonador e a vibração por simpatia das teclas. Contudo fiquei surpreso ao constatar que outro aspecto característico do som da kalimba havia sido capturado pela simulação da execução de notas rapidamente alternadas. Na frase apresentada no exemplo acima emergiu na superfície um ruído sutil e contínuo que resultou da combinação de transientes dos sucessivos impulsos gerados pelos ataques dos polegares virtuais. Esta resultante inesperada certamente contribuiu para recriar a aura especial da *m'bira* africana. Foi gratificante constatar que o modelo de síntese proposto, ainda que simples, foi capaz de simular aquela importante característica do instrumento tradicional.

Referências bibliográficas

AGORDOH, A. A. **African Music, Traditional and Contemporary**. New York: Nova Science Music. 2005.

BEAUCHAMP, J. "Analysis and Synthesis of Musical Instrument Sounds". In: **Analysis, Synthesis and Perception of Musical Sounds**. James W. Beauchamp (editor). New York: Springer. 2007.

De POLI, G. and ROCCHESSO, D. "Physically Based Sound Modelling." In: **Organised Sound** 3(1), Cambridge University Press, p. 61-76. 1998.

FLETCHER, N. and ROSSING, T. **The Physics of Musical Instruments**. New York: Springer-Verlag. 1998.

KLAPPER, M. "Working with Csound's adsyn, lpread and lpreson Opcodes." In: **The Csound Book**. Richard Boulanger (editor). Cambridge, MA: MIT Press. 2000.

NKETIA, J. H. K. **The Music of Africa**. New York: W. W. Norton. 1974.

SERRA, X. "Musical Sound Modeling with Sinusoids Plus Noise." In: **Musical Signal Processing**. Curtis Roads (editor). Lisse: Swets & Zeitlinger. 1997.

STEDMAN, J. G. "Voyages à Surinam et dans l'intérieur de la Guiane" (1777) apud FRUNGILO, M. D. **Dicionário de Percussão**. São Paulo: Editora da Unesp. 2002.

TRACEY, H. "Kalimba Magic" Disponível em: <<http://www.kalimbamagic.com/learnhow.html>>. Site visitado em março de 2010.

VÄLIMÄKI, V. and TAKALA, T. "Virtual Musical Instruments: Natural Sound Using Physical Models." **Organised Sound** 1(2), Cambridge University Press, p. 75-86. 1996.